
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 7.1.2.2.6 which is part of the ENDC test suite in the ‘iwd-TTCN3-B2018-12_D19wk23’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.2.2.6
Test Group:
\ENDC\7_1_2\
ATS Version:
iwd-TTCN3-B2018-12_D19wk23
System Simulator used:
Datang LinkTester 5G Protocol Conformance Test System
UE used:
MediaTek MT6297
Verification Status:
PASS


4. Corrections required for ENDC TC 7.1.2.2.6 using wk23
Change 4.1 

	Function name
	function f_TC_7_1_2_2_6_ENDC_NR ()

	Reason for change
	The v_T_Reassembly is set not the same as Table 7.1.2.2.6.3.1-1: RLC parameters.


	Summary of change
	Change v_T_Reassembly value to ms200.

	TTCN module
	RLC_ENDC_NR.ttcn

	MCC160 Comment
	Accepted


Before Change
       function f_TC_7_1_2_2_6_ENDC_NR() runs on ENDC_NR_PTC

  { // UM RLC / RLC re-establishment procedure

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList  v_DRB_ToAddModList;

    var SN_FieldLengthUM v_DL_Len;

    var SN_FieldLengthUM v_UL_Len;

    var T_Reassembly v_T_Reassembly;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    //Set null cipher Algorithm

    f_NR_AS_CipheringAlgorithm_Set(nea0); // @sic R5s190254 ch. 1.1 sic@

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    if (pc_um_WithShortSN) {

      v_DL_Len := size6;

      v_UL_Len := size6;

      v_RLC_Rec.UM_SN_Size := 6;

    } else {

      v_DL_Len := size12;

      v_UL_Len := size12;

      v_RLC_Rec.UM_SN_Size := 12;

    }
    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_UM_Test(tsc_NR_DRB2, v_DL_Len);

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config); // PDCP default is 18 bits from 38.508

    v_T_Reassembly := f_NR_DL_UM_T_Reassembly(f_NR_CellInfo_GetIsFR1(nr_Cell1));    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, v_DL_Len, v_UL_Len, v_T_Reassembly);

    // Get NR capabilities sent during EUTRA preamble

f_NR_ENDC_ConfigAM_UM(nr_Cell1, 0, 1, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList);
…
  }
After Change
     function f_TC_7_1_2_2_6_ENDC_NR() runs on ENDC_NR_PTC

  { // UM RLC / RLC re-establishment procedure

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList  v_DRB_ToAddModList;

    var SN_FieldLengthUM v_DL_Len;

    var SN_FieldLengthUM v_UL_Len;

    var T_Reassembly v_T_Reassembly;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    //Set null cipher Algorithm

    f_NR_AS_CipheringAlgorithm_Set(nea0); // @sic R5s190254 ch. 1.1 sic@

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    if (pc_um_WithShortSN) {

      v_DL_Len := size6;

      v_UL_Len := size6;

      v_RLC_Rec.UM_SN_Size := 6;

    } else {

      v_DL_Len := size12;

      v_UL_Len := size12;

      v_RLC_Rec.UM_SN_Size := 12;

    }
    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_UM_Test(tsc_NR_DRB2, v_DL_Len);

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config); // PDCP default is 18 bits from 38.508

v_T_Reassembly := ms200;    
v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, v_DL_Len, v_UL_Len, v_T_Reassembly);

    // Get NR capabilities sent during EUTRA preamble

f_NR_ENDC_ConfigAM_UM(nr_Cell1, 0, 1, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList);
…
  }
Change 4.2
	Function name
	function f_TC_7_1_2_2_6_NR_TestBody ()

	Reason for change
	1. UE will send first uplink data with MAC BSR (2 Bytes) at step 9, 80 bits size of UL grant is not suffient, need reduce v_RLC_Len1stSeg from 7 to 5.
2. When function fl_NR_GenerateRLC_UM_SDUs finished，PDCP SN of RLC SDU #3 and  #4 will be 2 and 3. The ack of dl data with PDCP SN = 1 has not been received after RRC reconfiguration in step 4, UE shall tansmit data with PDCP SN = 1, thus PDCP SN of RLC SDU #3 and #4 should be 1 and 2.

3. RLC SN length is size6 when pc_um_WithShortSN is true in step 4.

4. The 4th and 5th parameters of function f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant is p_Period  and p_Cycles .And in step 8A and 13A, 20 is period and 2 is cycle. So 2 and 20 shall be changed to 20 and 2 in the input parameter list.

5.  The test case should be test in UMD mode, so UMD receiving function shall be used.

	Summary of change
	1. Change 1st segment of RLC SDU size from 7 to 5 bytes according to the associated Prose CR.
2. Change PDCP SN of p_RLC_Rec.NR_RLC_DataList[2] and
p_RLC_Rec.NR_RLC_DataList[3] to 1 and 2.

3. Add branches according to pc_um_WithShortSN = true in step 4. 
4. Change the parameter period from 2 to 20, parameter cycle from 20 to 2 in step 8A and 13A.
5. Change function name from f_NR_RxAMD_PDUpartofSDU to f_NR_RxUMD_PDUpartofSDU and delete ? in inputting parameter list.
Associated Prose CR will be submitted in next RAN5 meeting.

	TTCN module
	RLC_TC_Common_NR.ttcn, RRCReconfig_ENDC_NR.ttcn

	MCC160 Comment
	1. Accepted
2. Accepted
3. Accepted

4. Accepted

5. Accepted


Before Change
         function f_TC_7_1_2_2_6_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                      DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { //UM RLC / RLC re-establishment procedure

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var integer v_RLC_Len1stSeg := 7; // Length of 1st segment
    var integer v_RLC_Len2ndSeg := 5; // Length of 2nd segment

    var integer v_RLC_SDUsize := v_RLC_Len1stSeg + v_RLC_Len2ndSeg;

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingDL2;

    var SubFrameTiming_Type v_TimingDL3;

    var SubFrameTiming_Type v_TimingUL;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    timer t_WatchDog;

    if (p_RLC_Rec.UM_SN_Size == 6) {

      // TBS for the first segment = RLC PDU data + RLC header without SO(1 octet) + MAC header(2 octets) = 7 + 1 + 2 = 10 octets => TBS = 80 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO(2 octets) + MAC header(2 octets) = 5 + 3 + 2 = 10 octets => TBS = 80 bits
      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(80, v_NR_UplinkBWP);

    } else {

      // TBS for the first segment = RLC PDU data + RLC header without SO + MAC header = 7 + 2 + 2 = 11 octets => TBS = 88 bits

      // TBS for the last segment = RLC PDU data + RLC header with SO + MAC header = 5 + 4 + 2 = 11 octets => TBS = 88 bits
      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);

    }

    fl_NR_GenerateRLC_UM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 4); // 4 SDUs starting from offset 0 [=index in DataList] tsc_RLC_SDU1 is at index 0 !

    // @siclog "Step 1" siclog@

    //The SS transmits UMD PDU#1. Header of UMD PDU#1 does not contain an SN. This PDU carries RLC SDU#1.

    f_NR_TxUMD_SDU(p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_Now, tsc_RLC_SDU1);

    // @siclog "Step 2" siclog@

    //The UE transmits RLC SDU#1.

    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1);

    // @siclog "Step 3" siclog@

    //The SS transmits UMD PDU#2. Header of UMD PDU#2 contains SN=0. This PDU carries the first segment of SDU#2.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 40); //40ms to give time for reconfiguration in step 4

    v_TimingDL2 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL1, 300); //(1.5 * t-Reassembly)

    v_TimingDL3 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL2, 20);

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU2, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU); // UM_TX_Next == 0

    // @siclog "Step 4" siclog@

    //The SS performs a RRCReconfiguration procedure including the secondaryCellGroup containing CellGroupConfig IE, including rlc-BearerToAddModList containing RLC-Bearer-Config for DRB with reestablishRLC set as true triggering RLC re-establishment.

    v_RLC_BearerConfigList[0] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_DRB(p_NR_DRB_Id), -, -, -, true_);

    f_NR_RLC_Reconfiguration(nr_Cell1, 0, 1, v_RLC_BearerConfigList); // UM_TX_Next == 0 and UM_RX_Next == 0
    // @siclog "Step 5" siclog@

    //The SS transmits UMD PDU#3. Header of UMD PDU#3 contains SN=0. This PDU carries the last segment of RLC SDU#2. The UE starts t- Reassembly.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU2, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 6" siclog@

    //Check: For 250 ms does the UE transmit RLC SDU#2?

    t_WatchDog.start(0.250);

    alt {

      [] a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU2) {f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");}

      [] t_WatchDog.timeout {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");}

    }

    // @siclog "Step 6A" siclog@

    //The SS stops allocating any UL grant

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    // @siclog "Step 7" siclog@

    //300 ms (1.5 * t- Reassembly) after step 5 the SS transmits UMD PDU#4. This PDU carries the first segment of RLC SDU#3.SN=1.

    p_RLC_Rec.UM_TX_Next := 1;  //Next SN=1

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL2), tsc_RLC_SDU3, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 8" siclog@

    //The SS transmits UMD PDU#5. This PDU carries the second and last segment of RLC SDU#3.SN=1.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL3), tsc_RLC_SDU3, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 8A" siclog@

    //SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#3 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL3, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 2, 20);
    // @siclog "Step 9" siclog@

    //Check: Does the UE transmit first segment of RLC SDU#3? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    // @siclog "Step 10" siclog@

    //Check: Does the UE transmit second and last  segment of RLC SDU#3? Header of UMD PDU contains SN=0.

    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3, ?, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    // @siclog "Step 11" siclog@

    //The SS performs a RRCReconfiguration procedure including the secondaryCellGroup containing CellGroupConfig IE, including rlc-BearerToAddModList containing RLC-Bearer-Config for DRB with reestablishRLC set as true triggering RLC re-establishment.

    //v_RLC_BearerConfigList as in step 4

    f_NR_RLC_Reconfiguration(nr_Cell1, 0, 1, v_RLC_BearerConfigList);

    p_RLC_Rec.UM_RX_Next := 0; //UE resets variables

    p_RLC_Rec.UM_TX_Next := 0; //UE resets variables

    // @siclog "Step 12" siclog@

    //The SS transmits UMD PDU#6. Header of UMD PDU#6 contains SN=0. This PDU carries the first segment of SDU#4.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU4, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 13" siclog@

    //The SS transmits UMD PDU#7. Header of UMD PDU#6 contains SN=0. This PDU carries the second segment of SDU#4.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU4, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 13A" siclog@

    //SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#4 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 2, 20);
    // @siclog "Step 14" siclog@

    //Check: Does the UE transmit first segment of RLC SDU#4? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    // @siclog "Step 15" siclog@

    //Check: Does the UE transmit second and last segment of RLC SDU#4? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

  }
After Change
   function f_TC_7_1_2_2_6_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                      DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { //UM RLC / RLC re-establishment procedure

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var integer v_RLC_Len1stSeg := 5; // Length of 1st segment
    var integer v_RLC_Len2ndSeg := 5; // Length of 2nd segment

    var integer v_RLC_SDUsize := v_RLC_Len1stSeg + v_RLC_Len2ndSeg;

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingDL1;

    var SubFrameTiming_Type v_TimingDL2;

    var SubFrameTiming_Type v_TimingDL3;

    var SubFrameTiming_Type v_TimingUL;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    timer t_WatchDog;

    if (p_RLC_Rec.UM_SN_Size == 6) {

      // TBS for the first segment = RLC SDU data + RLC header without SO(1 octet) + MAC BSR(2 octets) + MAC header(2 octets) = 5 + 1 + 2 + 2 = 10 octets => TBS = 80 bits

      // TBS for the last segment = RLC SDU data + RLC header with SO(3 octets) + MAC header(2 octets) = 5 + 3 + 2 = 10 octets => TBS = 80 bits
      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(80, v_NR_UplinkBWP);

    } else {

      // TBS for the first segment = RLC SDU data + RLC header without SO + MAC BSR(2 octets) + MAC header = 5 + 2 + 2 + 2 = 11 octets => TBS = 88 bits

      // TBS for the last segment = RLC SDU data + RLC header with SO + MAC header = 5 + 4 + 2 = 11 octets => TBS = 88 bits
      v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(88, v_NR_UplinkBWP);

    }

fl_NR_GenerateRLC_UM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 2);  
p_RLC_Rec. PDCP_TX_Next := 1;

fl_NR_GenerateRLC_UM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 2, 2);     
    // @siclog "Step 1" siclog@

    //The SS transmits UMD PDU#1. Header of UMD PDU#1 does not contain an SN. This PDU carries RLC SDU#1.

    f_NR_TxUMD_SDU(p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_Now, tsc_RLC_SDU1);

    // @siclog "Step 2" siclog@

    //The UE transmits RLC SDU#1.

    a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1);

    // @siclog "Step 3" siclog@

    //The SS transmits UMD PDU#2. Header of UMD PDU#2 contains SN=0. This PDU carries the first segment of SDU#2.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 40); //40ms to give time for reconfiguration in step 4

    v_TimingDL2 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL1, 300); //(1.5 * t-Reassembly)

    v_TimingDL3 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL2, 20); 

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU2, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU); // UM_TX_Next == 0

    // @siclog "Step 4" siclog@

    //The SS performs a RRCReconfiguration procedure including the secondaryCellGroup containing CellGroupConfig IE, including rlc-BearerToAddModList containing RLC-Bearer-Config for DRB with reestablishRLC set as true triggering RLC re-establishment.
    if(pc_um_WithShortSN)

    {

      v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, size6, size6, ms200, true_);

    }

    else

    {

      v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, -, -, ms200, true_);

    }
     f_NR_RLC_Reconfiguration(nr_Cell1, 0, 1, v_RLC_BearerConfigList); // UM_TX_Next == 0 and UM_RX_Next == 0


    // @siclog "Step 5" siclog@

    //The SS transmits UMD PDU#3. Header of UMD PDU#3 contains SN=0. This PDU carries the last segment of RLC SDU#2. The UE starts t- Reassembly.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU2, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 6" siclog@

    //Check: For 250 ms does the UE transmit RLC SDU#2?

    t_WatchDog.start(0.250);

    alt {

      [] a_NR_RxUMD_oneSDU(p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU2) {f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");}

      [] t_WatchDog.timeout {f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");}

    }

    // @siclog "Step 6A" siclog@

    //The SS stops allocating any UL grant

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    // @siclog "Step 7" siclog@

    //300 ms (1.5 * t- Reassembly) after step 5 the SS transmits UMD PDU#4. This PDU carries the first segment of RLC SDU#3.SN=1.

    p_RLC_Rec.UM_TX_Next := 1;  //Next SN=1

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL2), tsc_RLC_SDU3, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 8" siclog@

    //The SS transmits UMD PDU#5. This PDU carries the second and last segment of RLC SDU#3.SN=1.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL3), tsc_RLC_SDU3, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 8A" siclog@

    //SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#3 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL3, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);
    // @siclog "Step 9" siclog@

    //Check: Does the UE transmit first segment of RLC SDU#3? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    // @siclog "Step 10" siclog@

    //Check: Does the UE transmit second and last  segment of RLC SDU#3? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3, /*?,*/ v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 10");

    // @siclog "Step 11" siclog@

    //The SS performs a RRCReconfiguration procedure including the secondaryCellGroup containing CellGroupConfig IE, including rlc-BearerToAddModList containing RLC-Bearer-Config for DRB with reestablishRLC set as true triggering RLC re-establishment.

    //v_RLC_BearerConfigList as in step 4

    f_NR_RLC_Reconfiguration(nr_Cell1, 0, 1, v_RLC_BearerConfigList);

    p_RLC_Rec.UM_RX_Next := 0; //UE resets variables

    p_RLC_Rec.UM_TX_Next := 0; //UE resets variables

    // @siclog "Step 12" siclog@

    //The SS transmits UMD PDU#6. Header of UMD PDU#6 contains SN=0. This PDU carries the first segment of SDU#4.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of send the first segment

    v_TimingDL1 := f_SubFrameTiming_AddMilliSeconds(v_TimingDL, 20);

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL), tsc_RLC_SDU4, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    // @siclog "Step 13" siclog@

    //The SS transmits UMD PDU#7. Header of UMD PDU#6 contains SN=0. This PDU carries the second segment of SDU#4.

    f_NR_TxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, cs_TimingInfo_NR(v_TimingDL1), tsc_RLC_SDU4, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    // @siclog "Step 13A" siclog@

    //SS allocates 2 UL grants at an interval of 20 ms so as to loop back RLC SDU#4 in 2 RLC/MAC PDUs.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL1, 20);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 2);
    // @siclog "Step 14" siclog@

    //Check: Does the UE transmit first segment of RLC SDU#4? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    // @siclog "Step 15" siclog@

    //Check: Does the UE transmit second and last segment of RLC SDU#4? Header of UMD PDU contains SN=0.

    f_NR_RxUMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

  }
5. Execution Log Files
5.1 MediaTek MT6297
The MediaTek MT6297 UE passed this test case on Datang LinkTester 5G Protocol Conformance Test System on EN-DC band combination 3A-N78. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_7_1_2_2_6_ENDC_3A_N78_LOG.log
References

	[1]
	R5s190538:   Supporting information for agreement of EN-DC test case 7.1.2.2.6 in FR1
                          This archive comprises:

                        -  text and xml format execution log files 


